
[bookmark: _Toc39053579]5.1.1.11	Built-in classes
The abstract special built-in class called object is the superclass for all classes that do not explicitly extend another class.
The pseudo definition of that class is:
type class @abstract @builtin object {
        // This function will return a tool-specific descriptive string by default
        // but can be overridden by subclasses
  		public function toString() return universal charstring;

		// Indicates wether some object is semantically equivalent to this one,
		//  according to some equivalence relation.
		// The default implementation returns true if and only if both this and obj
		//  are the same object instance, otherwise returns false.
		public function equals(object obj) return boolean {
			return this == obj;
		}

}

NOTE:	The @builtin is only added for illustrative purposes and not part of the TTCN-3 language.
[bookmark: _Toc39053580]5.1.1.12	Nested classes
A class type definition may occur also as a member of another class type definition body. Such a class is called a nested class while the surrounding class is called the containing class.
Members defined in the body of a nested class may access all named entities that are accessible in the scope of the containing class with the same restrictions.
If a nested class does not have a runs on clause it inherits the runs on type from its enclosing class.
If a nested class does not have a system clause it inherits the system type from its enclosing class.
If a nested class does not have an mtc clause it inherits the mtc type from its enclosing class.
The type of the nested class may be referenced with the dotted notation applied to a type reference of the enclosing class.
The constructor of a nested class may be invoked on a reference composed of an instance of the containing class followed by a dot and nested class identifier. Inside the scope of the containing class, the identifier of the nested class may be used without dotted notation for the use of calling its constructor.
Restrictions
1. The members of a nested class shall not have the same name as one of the members of a (directly or indirectly) containing class.
1. Referencing the name of a nested class in a null reference via dotted notation shall cause an error.
EXAMPLE:
type record of charstring Strings;

type class @abstract StringIterator {
  function @abstract hasNext() return boolean;
  function @abstract next() return charstring;
}

type class StringList {
  var Strings v_strings;
  
  type class Iterator extends StringIterator {
     var integer v_pos := 0;
     
     public function hasNext() return boolean {
       return v_pos < lengthof(v_strings);
     }

     public function next() return charstring {
       v_pos := v_pos + 1;
       return v_strings[v_pos-1];
     }
  }

  function iterator() return Iterator {
    return Iterator.create();
  }
}
var StringList v_list := StringList.create();
var StringList.Iterator v_iterator := v_list.Iterator.create();
v_list := null;
v_iterator := v_list.Iterator.create(); // error

EXAMPLE:
type class Square {
…
  public function equals(object obj) return boolean {
     if (obj of Rectangle) {
		// a rectangle is a suare if it has 4 sides of equal lengths
		 var Rectangle rectangle := obj => Rectangle;
		 if (rectangle.getNofSides() != 4) {
			 return false;
		 }
		 var integer tempSideLength := rectangle.getSideLength(0);
		 return tempSideLength == rectangle.getSideLength(1) and tempSideLength == rectangle.getSideLength(2) and tempSideLength == rectangle.getSideLength(3);
	 }

	 return this == obj;
  }
}
type class Rectangle {
…
  public function getNofSides() return integer { … }
  public function getSideLength(in integer index) return integer { … }
  public function equals(object obj) return boolean {
     if (obj of Square) {
			 // a square is always a rectangle
			 return true;
		 }

		 return this == obj;
  }
}

